

## Εργαστηριακές ασκήσεις στον Μικροελεγκτή AVR

### 3<sup>η</sup> Εργαστηριακή Άσκηση του AVR – Χρήση πληκτρολογίου και οθόνη

Στην άσκηση αυτή θα γίνει μελέτη της χρήσης του πληκτρολογίου 4×4 και της αλφαριθμητικής οθόνης LCD 2×16 χαρακτήρων (επικοινωνία μεταξύ οθόνης και του μικροελεγκτή γίνεται με λέξεις των 4 bit).


#### Έλεγχος Περιφερειακών Συσκευών

Η επικοινωνία ενός μικροελεγκτή με περιφερειακές συσκευές μπορεί να γίνει με διαφορετικούς τρόπους και πρωτόκολλα. Για κάθε διαφορετική περίπτωση και τεχνική επικοινωνίας, είναι ιδιαίτερα χρήσιμη η κατασκευή μιας βιβλιοθήκης λογισμικού (ρουτίνες) που αναλαμβάνουν όλες τις χαμηλού επιπέδου λειτουργίες (π.χ. τροφοδοσία και έλεγχος συγκεκριμένων ακροδεκτών, χρονισμός) που καθορίζονται από τον τρόπο επικοινωνίας, και παρέχουν υψηλότερου επιπέδου λειτουργίες (π.χ. αποστολή και λήψη χαρακτήρων), που χρησιμοποιούνται για την κατασκευή λογισμικού εφαρμογών. Οι βιβλιοθήκες αυτές ονομάζονται οδηγοί συσκευών. Για την κατασκευή τους είναι απαραίτητη η γνώση του τρόπου επικοινωνίας και της συνδεσμολογίας της περιφερειακής συσκευής. Συχνά, η περιφερειακή συσκευή συνοδεύεται από έναν ελεγκτή ο οποίος καθορίζει και υλοποιεί τον τρόπο επικοινωνίας. Τα τεχνικά του χαρακτηριστικά περιέχονται στο αντίστοιχο εγχειρίδιο τεχνικών προδιαγραφών. Η συνδεσμολογία προκύπτει από τα τεχνικά χαρακτηριστικά αλλά και τους περιορισμούς του συστήματος που θα συνδεθεί η περιφερειακή συσκευή (π.χ. διαθεσιμότητα ακροδεκτών).

Στην αναπτυξιακή πλακέτα EasyAVR6 διαθέτει μεταξύ των άλλων περιφερειακών συσκευών ένα πληκτρολόγιο 4×4 και μια οθόνη χαρακτήρων 2×16. Στην συνέχεια παρουσιάζονται οι συσκευές αυτές σε συνδυασμό με το αντίστοιχο λογισμικό οδήγησής τους.

#### Πληκτρολόγιο 4×4

Το πληκτρολόγιο 4×4 του αναπτυξιακού easyAVR6 αποτελείται από 16 πιεστικούς διακόπτες συνδεδεμένους όπως φαίνεται στο ακόλουθο σχήμα.



Σχήμα 3.1 Το πληκτρολόγιο 4×4 του αναπτυξιακού easyAVR6

Η ανάγνωση του πληκτρολογίου γίνεται ανά γραμμή. Η γραμμή που επιθυμούμε να ελέγξουμε για πατημένους διακόπτες επιλέγεται θέτοντας έναν από τους ακροδέκτες PC7 – PC4 της θύρας PORTC του μικροελεγκτή στο λογικό 1. Οι 4 γραμμές του πληκτρολογίου επιλέγονται από τους ακροδέκτες PC7 – PC4 οι οποίοι πρέπει να είναι ρυθμισμένοι για έξοδο. Στη συνέχεια εντοπίζουμε τις στήλες των πιεσμένων διακοπών, διαβάζοντας τον καταχωρητή PINC και ελέγχοντας τους ακροδέκτες PC3 – PC0 που τους έχουμε ρυθμίσει ως εισόδους. Λογικό 1 αντιστοιχεί σε πιεσμένο διακόπτη. Αν δεν υπάρχει πιεσμένος διακόπτης, λόγω των pull-down αντιστάσεων, έχουμε λογικό 0 στην είσοδο.

Η αρχικοποίηση της θύρα PORTC του μικροελεγκτή μπορεί να πραγματοποιηθεί με τις ακόλουθες εντολές (οι εσωτερικές αντιστάσεις pull-up πρέπει να είναι απενεργοποιημένες).

```
ldi r24 ,(1 << PC7) | (1 << PC6) | (1 << PC5) | (1 << PC4) ; θέτει ως εξόδους τα 4 MSB
out DDRC ,r24 ; της θύρας PORTC
```

Για να ελέγξουμε μια γραμμή του πληκτρολογίου για πιεσμένους διακόπτες μπορούμε να χρησιμοποιήσουμε τη ρουτίνα `scan_row`. Στον καταχωρητή r24 αποθηκεύουμε τον αριθμό της γραμμής (1 – 4) που θέλουμε να ελέγξουμε και ανακτούμε το αποτέλεσμα του ελέγχου στα 4 λιγότερα σημαντικά bit του ίδιου καταχωρητή. Οι δύο εντολές `nop` μεταξύ της εντολής που επιλέγει τη γραμμή και της εντολής που διαβάζει τον καταχωρητή PINC είναι απαραίτητες καθώς μπορεί να απαιτηθούν μέχρι και δύο κύκλοι ρολογιού μέχρι μια αλλαγή στην κατάσταση των ακροδεκτών (λόγω της αδράνειας τους) να καταγραφεί στον καταχωρητή PINC.

#### Ρουτίνα: `scan_row`

Έλεγχος μιας γραμμής του πληκτρολογίου για πιεσμένους διακόπτες.

**Είσοδος:** Ο αριθμός της γραμμής προς ανάγνωση πρέπει να είναι αποθηκευμένος στον καταχωρητή r24 (τιμή 1-4).

**Έξοδος:** Στα 4 λιγότερο σημαντικά bit του r24 είναι αποθηκευμένη η κατάσταση κάθε διακόπτη.

**Καταχωρητές:** r25:r24

**Καλούμενες υπορουτίνες:** -

```
scan_row:
    ldi r25 ,0x08 ; αρχικοποίηση με '0000 1000'
back_:   lsl r25 ; αριστερή ολίσθηση του '1' τόσες θέσεις
        dec r24 ; όσος είναι ο αριθμός της γραμμής
        brne back_
        out PORTC ,r25 ; η αντίστοιχη γραμμή τίθεται στο λογικό '1'
        nop
        nop ; καθυστέρηση για να προλάβει να γίνει η αλλαγή κατάστασης
        in r24 ,PINC ; επιστρέφουν οι θέσεις (στήλες) των διακοπών που είναι πιεσμένοι
        andi r24 ,0x0f ; απομονώνονται τα 4 LSB όπου τα '1' δείχνουν που είναι πατημένοι
        ret ; οι διακόπτες.
```

Τώρα που διαθέτουμε μια ρουτίνα που ελέγχει μια γραμμή του πληκτρολογίου για πιεσμένους διακόπτες μπορούμε να γράψουμε μια άλλη που θα την ενσωματώνει και θα ελέγχει ολόκληρο το πληκτρολόγιο. Το αποτέλεσμα του ελέγχου είναι ένας δυαδικός αριθμός με μήκος 16 bit που αποθηκεύεται στους καταχωρητές r25:r24. Κάθε διακόπτης από τους 16 αντιστοιχεί σε ένα bit των καταχωρητών r25:r24 όπου λογικό '1' σημαίνει ότι είναι πιεσμένος ο αντίστοιχος διακόπτης. Η αντιστοιχία φαίνεται στο επόμενο σχήμα.

r25								r24							
A	3	2	1	B	6	5	4	C	9	8	7	D	#	0	*

**Ρουτίνα: scan\_keypad**

Έλεγχος του πληκτρολογίου για πιεσμένους διακόπτες.

**Είσοδος:** καμία

**Έξοδος:** Στους καταχωρητές r25:r24 είναι αποθηκευμένη η κατάσταση των 16 διακοπών του πληκτρολογίου.

**Καταχωρητές:** r27:r26, r25:r24

**Καλούμενες υπορουτίνες:** scan\_row

scan\_keypad:

```
ldi r24,0x01          ; έλεγξε την πρώτη γραμμή του πληκτρολογίου
rcall scan_row
swap r24              ; αποθήκευσε το αποτέλεσμα
mov r27,r24           ; στα 4 msb του r27
ldi r24,0x02          ; έλεγξε τη δεύτερη γραμμή του πληκτρολογίου
rcall scan_row
add r27,r24           ; αποθήκευσε το αποτέλεσμα στα 4 lsb του r27
ldi r24,0x03          ; έλεγξε την τρίτη γραμμή του πληκτρολογίου
rcall scan_row
swap r24              ; αποθήκευσε το αποτέλεσμα
mov r26,r24           ; στα 4 msb του r26
ldi r24,0x04          ; έλεγξε την τέταρτη γραμμή του πληκτρολογίου
rcall scan_row
add r26,r24           ; αποθήκευσε το αποτέλεσμα στα 4 lsb του r26
movw r24,r26          ; μετέφερε το αποτέλεσμα στους καταχωρητές r25:r24
ret
```

Έως τώρα έχουμε δει πώς να αρχικοποιήσουμε την θύρα PORTC του μικροελεγκτή για να χρησιμοποιήσουμε το πληκτρολόγιο και πώς να το ελέγξουμε για πιεσμένα πλήκτρα. Όμως αυτό που πραγματικά μας ενδιαφέρει δεν είναι ποιοι διακόπτες είναι πιεσμένοι τη στιγμή που ελέγχουμε το πληκτρολόγιο, αλλά ποιο πλήκτρο πάτησε ο χρήστης. Όταν ο χρήστης πατάει ένα πλήκτρο, αυτό μπορεί να μείνει πιεσμένο για αυθαίρετα μεγάλο χρονικό διάστημα. **Πρέπει να μπορούμε να ξεχωρίσουμε ποιο πλήκτρο πατήθηκε (ως ολοκληρωμένη ενέργεια) και όχι ποιο πλήκτρο είναι πατημένο.**

Για να το επιτύχουμε πρέπει να ελέγχουμε το πληκτρολόγιο για πιεσμένους διακόπτες με την ρουτίνα scan\_keypad, **να αποθηκεύουμε το αποτέλεσμα στη μνήμη RAM του μικροελεγκτή** και στη συνέχεια με μια δεύτερη κλήση της ίδιας ρουτίνας να ξαναελέγξουμε το πληκτρολόγιο. Μια σύγκριση των δύο καταστάσεων του πληκτρολογίου θα αποκαλύψει τις διαφορές στην κατάσταση των διακοπών. Το χρονικό διάστημα ανάμεσα στις διαδοχικές κλήσεις της συνάρτησης scan\_keypad είναι κρίσιμο, διότι καθορίζει το χρόνο που θα πρέπει να μείνει πιεσμένος ένας διακόπτης από το χρήστη για να καταγραφεί από τον μικροελεγκτή. Αυτό σημαίνει ότι ένα μεγάλο χρονικό διάστημα μεταξύ των διαδοχικών κλήσεων της ρουτίνας θα αναγκάσει το χρήστη να κρατά πατημένο το πλήκτρο για αντίστοιχα μεγάλο χρονικό διάστημα (ώστε να μπορεί να αναγνωριστεί). Αντίθετα, ένα πολύ μικρό χρονικό διάστημα θα δημιουργήσει προβλήματα λόγω του σπινθηρισμού που παρουσιάζουν οι διακόπτες.

Η ρουτίνα scan\_keypad\_rising\_edge υλοποιεί όσα αναφέρθηκαν προωύτερα και παράλληλα αντιμετωπίζει αποτελεσματικά το ζήτημα του σπινθηρισμού των διακοπών.

**Ρουτίνα: scan\_keypad\_rising\_edge**

Έλεγχος του πληκτρολογίου για διακόπτες που δεν ήταν πιεσμένοι την τελευταία φορά που κλήθηκε η ρουτίνα και τώρα είναι.

**Είσοδος:** Ο αναμενόμενος χρόνος σπινθηρισμού των διακοπών σε ms είναι αποθηκευμένος στον καταχωρητή r24.

**Έξοδος:** Ένας αριθμός των 16 bit, ενδεικτικός των διακοπών που «μόλις» πατήθηκαν, είναι αποθηκευμένος στους καταχωρητές r25:r24

**Καταχωρητές:** r27:r26, r25:r24, r23:r22

**Καλούμενες υπορουτίνες:** scan\_keypad, wait\_msec

**Παρατηρήσεις:**

Για να καταγραφούν οι διακόπτες που έχουν «μόλις» πιεστεί, η ρουτίνα χρησιμοποιεί την μεταβλητή `_tmp_`. Επειδή η μεταβλητή `_tmp_` βρίσκεται στην RAM του μικροελεγκτή δεν είναι αρχικοποιημένη. Ο χρήστης πρέπει να την αρχικοποιήσει είτε ρητά είτε με μια κλήση της ρουτίνας μόνο για αυτό το σκοπό.

; ---- Αρχή τμήματος δεδομένων

.DSEG

`_tmp_` : .byte 2

; ---- Τέλος τμήματος δεδομένων

.CSEG

```
scan_keypad_rising_edge:
    mov r22 ,r24          ; αποθήκευσε το χρόνο σπινθηρισμού στον r22
    rcall scan_keypad     ; έλεγξε το πληκτρολόγιο για πιεσμένους διακόπτες
    push r24              ; και αποθήκευσε το αποτέλεσμα
    push r25
    mov r24 ,r22          ; καθυστέρησε r22 ms (τυπικές τιμές 10-20 msec που καθορίζεται από τον
    ldi r25 ,0             ; κατασκευαστή του πληκτρολογίου – χρονοδιάρκεια σπινθηρισμών)
    rcall wait_msec
    rcall scan_keypad     ; έλεγξε το πληκτρολόγιο ξανά και
    pop r23               ; απόρριψε όσα πλήκτρα εμφανίζουν
    pop r22               ; σπινθηρισμό
    and r24 ,r22
    and r25 ,r23
    ldi r26 ,low(_tmp_)   ; φόρτωσε την κατάσταση των διακοπών στην
    ldi r27 ,high(_tmp_)  ; προηγούμενη κλήση της ρουτίνας στους r27:r26
    ld r23 ,X+
    ld r22 ,X
    st X ,r24             ; αποθήκευσε στη RAM τη νέα κατάσταση
    st -X ,r25            ; των διακοπών
    com r23
    com r22               ; βρες τους διακόπτες που έχουν «μόλις» πατηθεί
    and r24 ,r22
    and r25 ,r23
    ret
```

Τέλος, επειδή η κατάσταση του πληκτρολογίου στην μορφή ενός δυαδικού αριθμού δεν είναι ιδιαίτερα χρήσιμη υπάρχει και η ρουτίνα `keypad_to_ascii` που εντοπίζει τον διακόπτη που έχει πατηθεί και επιστρέφει τον κωδικό `ascii` του χαρακτήρα που αντιστοιχεί στον διακόπτη. Αν δεν είναι πιεσμένος κανένας διακόπτης επιστρέφει την τιμή 0, ενώ εάν είναι πατημένοι πολλοί επιστρέφει μόνο έναν από αυτούς (ο 1<sup>ος</sup> που εντοπίζεται με βάση την σειρά εξερεύνησης των εντολών της ρουτίνας που ακολουθεί).

**Ρουτίνα: keypad\_to\_ascii**

Αντιστοίχιση διακοπών, κωδικών `ascii`.

**Είσοδος:** Στους καταχωρητές `r25:r24` είναι αποθηκευμένος ένας αριθμός 16 bit, ενδεικτικός της κατάστασης κάθε διακόπτη.

**Έξοδος:** Ο κωδικός `ascii`, που αντιστοιχεί στον πρώτο πατημένο διακόπτη που εντοπίστηκε, αποθηκεύεται στον καταχωρητή `r24` ή 0 αν δεν έχει πατηθεί κάποιος.

**Καταχωρητές:** `r27:r26`, `r25:r24`

**Παρατηρήσεις:** Ο αριθμός των 16 bit που αποθηκεύεται στους καταχωρητές `r25:r24` κατά την κλήση της ρουτίνας πρέπει να προέρχεται από μια εκ των `scan_keypad` ή `scan_keypad_rising_edge`.

```

keypad_to_ascii:
    movw r26 ,r24
    ldi r24 ,'*'
    sbrc r26 ,0
    ret
    ldi r24 , '0'
    sbrc r26 ,1
    ret
    ldi r24 , '#'
    sbrc r26 ,2
    ret
    ldi r24 , 'D'
    sbrc r26 ,3
    ret
    ldi r24 , '7'
    sbrc r26 ,4
    ret
    ldi r24 , '8'
    sbrc r26 ,5
    ret
    ldi r24 , '9'
    sbrc r26 ,6
    ret
    ldi r24 , 'C'
    sbrc r26 ,7
    ret
    ldi r24 , '4'
    sbrc r27 ,0
    ret
    ldi r24 , '5'
    sbrc r27 ,1
    ret
    ldi r24 , '6'
    sbrc r27 ,2
    ret
    ldi r24 , 'B'
    sbrc r27 ,3
    ret
    ldi r24 , '1'
    sbrc r27 ,4
    ret
    ldi r24 , '2'
    sbrc r27 ,5
    ret
    ldi r24 , '3'
    sbrc r27 ,6
    ret
    ldi r24 , 'A'
    sbrc r27 ,7
    ret
    clr r24
    ret

```

; λογικό '1' στις θέσεις του καταχωρητή r26 δηλώνουν  
; τα παρακάτω σύμβολα και αριθμούς

r26

C	9	8	7	D	#	0	*
---	---	---	---	---	---	---	---

; αν δεν είναι '1' παρακάμπτει την ret, αλλιώς (αν είναι '1')  
; επιστρέφει με τον καταχωρητή r24 την ASCII τιμή του D.


; λογικό '1' στις θέσεις του καταχωρητή r27 δηλώνουν  
; τα παρακάτω σύμβολα και αριθμούς

r27

A	3	2	1	B	6	5	4
---	---	---	---	---	---	---	---


### Αλφαριθμητική Οθόνη Χαρακτήρων 2×16

Στην αναπτυξιακή πλακέτα EasyAVR6 μια πολύ χρήσιμη περιφερειακή συσκευή είναι η πρόσθετη οθόνη χαρακτήρων 2×16. Η συνδεσμολογία γίνεται με τους 6 περισσότερο σημαντικούς ακροδέκτες της θύρας PORTD (PD2-PD7), όπως φαίνεται στο σχήμα 6.1. Οι ακροδέκτες αυτοί συνδέονται σε ακροδέκτες του ελεγκτή της οθόνης WH1602B, του οποίου το εγχειρίδιο τεχνικών προδιαγραφών παρατίθεται στις επόμενες σελίδες. Από τους 6 ακροδέκτες, οι 2 λιγότερο σημαντικοί (PD2 και PD3) είναι ακροδέκτες ελέγχου ενώ οι υπόλοιποι 4 σχηματίζουν λέξεις των 4 bit με τις οποίες γίνεται η επικοινωνία μικροελεγκτή – ελεγκτή οθόνης.




Σχήμα 3.2. Κυκλωματικό διάγραμμα οθόνης χαρακτήρων 2×16 στην αναπτυξιακή πλακέτα EasyAVR6.

Το αναπτυξιακό easyAVR6 διαθέτει μια αλφαριθμητική οθόνη χαρακτήρων. Για να χρησιμοποιήσουμε το συγκεκριμένο περιφερειακό είναι απαραίτητο να εξοικειωθούμε με τη δομή και τη λειτουργία του. Η εσωτερική οργάνωση της αλφαριθμητικής οθόνης χαρακτήρων παρουσιάζεται στο ακόλουθο σχήμα.



Σχήμα 3.3. Διάγραμμα διασύνδεσης του ελεγκτή HD44780 με LSD οθόνη χαρακτήρων 2×16

Κεντρικό ρόλο παίζει ο ελεγκτής HD44780, ο οποίος αναλαμβάνει να παρουσιάσει στην οθόνη υγρών κρυστάλλων τους αλφαριθμητικούς χαρακτήρες καθώς και την αλληλεπίδραση με τον χρήστη. Το μπλοκ διάγραμμα του ελεγκτή HD44780 φαίνεται στο παρακάτω σχήμα.



Σχήμα 3.4. Διάγραμμα βαθμίδων του ελεγκτή οθόνης LSD HD44780.

### Περιγραφή λειτουργία

Η μονάδα οθόνης LCD περιλαμβάνει ελεγκτή που έχει δύο 8-bit καταχωρητές, έναν καταχωρητή εντολών (instruction register -IR) και έναν καταχωρητή δεδομένων (data register- DR).

### Καταχωρητές

Ο καταχωρητής IR αποθηκεύει κωδικούς εντολών, όπως καθαρισμός οθόνης ολίσθηση δρομέα, και πληροφορίες για τη διεύθυνση των δεδομένων της RAM απεικόνισης (display data RAM -DDRAM) και για τη γεννήτρια χαρακτήρων RAM (character generator RAM- CGRAM). Ο καταχωρητής IR εγγράφεται μόνο από τον Μικροελεγκτή. Ο καταχωρητής DR αποθηκεύει προσωρινά τα δεδομένα για να εγγραφούν σε DDRAM ή CGRAM και προσωρινά αποθηκεύει δεδομένα που πρέπει να διαβαστούν από την DDRAM ή την CGRAM. Τα δεδομένα γράφονται στον DR από τον Μικροελεγκτή αυτόματα μεταφέρονται σε DDRAM ή CGRAM από μια εσωτερική λειτουργία. Το DR χρησιμοποιείται επίσης για αποθήκευση κατά την ανάγνωση δεδομένων από DDRAM ή CGRAM. Όταν πληροφορίες διεύθυνσης εγγράφονται στο IR, τα αντίστοιχα δεδομένα αποθηκεύονται στον DR από την DDRAM ή την CGRAM μέσω μιας εσωτερικής λειτουργίας. Η μεταφορά δεδομένων στον MPU ολοκληρώνεται όταν διαβάσει τον DR. Μετά την ανάγνωση, τα δεδομένα των DDRAM ή CGRAM της επόμενης διεύθυνσης αποστέλλονται στον DR για την επόμενη ανάγνωση από

τον Μικροελεγκτή. Με το σήμα επιλογής καταχωρητή (register selector -RS), οι δύο αυτοί καταχωρητές μπορούν να επιλεγούν (Πίνακας 1).

#### Σημαία απασχόλησης (Busy Flag-BF)

Όταν η σημαία απασχόλησης είναι 1, η HD44780U είναι σε κατάσταση εσωτερικής λειτουργίας, και επόμενη εντολή δεν γίνεται δεκτή. Όταν RS = 0 και R / W = 1 (Πίνακας 1), η σημαία απασχόλησης εμφανίζεται στην έξοδο DB7. Η επόμενη εντολή πρέπει να είναι δοθεί, μετά την εξασφάλιση ότι η σημαία απασχόλησης είναι 0.

#### Μετρητής Διεύθυνση (Address Counter -AC)

Ο μετρητής διεύθυνση (AC) παρέχει διευθύνσεις σε αμφότερες τις μνήμες DDRAM και CGRAM. Όταν μια διεύθυνση μιας εντολής εγγράφεται στον IR, η διεύθυνση στέλνεται από τον IR στον AC. Η επιλογή είτε της DDRAM ή της CGRAM καθορίζεται επίσης από την εντολή.

Μετά την εγγραφή προς (ή την ανάγνωση από) τις DDRAM ή CGRAM, η AC αυτόματα αυξάνεται κατά 1 (ή ελαττώνεται κατά 1). Το περιεχόμενο του καταχωρητή AC δίνεται στη έξοδο μέσω των DB0 - DB6 όταν RS = 0 και R/W = 1 (βλέπε Πίνακα 1).

Πίνακας 1. Επιλογή Καταχωρητή

RS	R/W	Operation
0	0	IR write as an internal operation (display clear, etc.)
0	1	Read busy flag (DB7) and address counter (DB0 to DB6)
1	0	DR write as an internal operation (DR to DDRAM or CGRAM)
1	1	DR read as an internal operation (DDRAM or CGRAM to DR)

### Περιγραφή των εσωτερικών δομικών μονάδων του ελεγκτή

#### Μνήμες

Ο ελεγκτής διαθέτει συνολικά 3 μνήμες, δύο τύπου RAM και μια τύπου ROM. Η ROM δημιουργίας χαρακτήρων χρησιμοποιείται για τη δημιουργία εικόνων χαρακτήρων 5×8 κουκίδων ή 5×10 κουκίδων από κωδικούς χαρακτήρων των 8 bit. Η αντιστοιχία αριθμών των 8 bit και χαρακτήρων φαίνεται στον ακόλουθο πίνακα.



Upper 4 bit Lower 4 bit	LLLL	LLLH	LLHL	LLHH	LHLL	LHLH	LHHL	LHHH	HLLL	HLLH	HLHL	HLHH	HHLL	HHLH	HHHL	HHHH
LLLL	CG RAM (1)	!			0	1	2	3	4	5	6	7	8	9	A	B
LLLH	CG RAM (2)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LLHL	CG RAM (3)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LLHH	CG RAM (4)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LHLL	CG RAM (5)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LHLH	CG RAM (6)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LHHL	CG RAM (7)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
LHHH	CG RAM (8)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HLLL	CG RAM (1)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HLLH	CG RAM (2)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HLHL	CG RAM (3)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HLHH	CG RAM (4)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HHLL	CG RAM (5)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HHLH	CG RAM (6)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HHHL	CG RAM (7)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C
HHHH	CG RAM (8)	!	!	!	1	2	3	4	5	6	7	8	9	A	B	C

Η RAM δημιουργίας χαρακτήρων (character generator RAM- CGRAM) εκτελεί την ίδια λειτουργία με την ROM δημιουργίας χαρακτήρων, με την διαφορά ότι οι χαρακτήρες που απεικονίζονται πρέπει να δημιουργηθούν από τον χρήστη και να αποθηκευτούν σε αυτή.

Η RAM απεικόνισης δεδομένων (display data RAM –DDRAM) αποθηκεύει τα δεδομένα που πρόκειται να απεικονιστούν στην οθόνη. Η χωρητικότητά της είναι 80 byte. Η αντιστοιχία μεταξύ θέσεων στην RAM απεικόνισης χαρακτήρων και θέσεων στην οθόνη φαίνεται στην ακόλουθη εικόνα. Παρατηρήστε ότι αρκετές θέσεις της μνήμης δεν απεικονίζονται στην οθόνη και πως όταν η οθόνη έχει διαμόρφωση για απεικόνιση



— Case 2: For a 16-character × 2-line display, the HD44780 can be extended using one 40-output extension driver.

When display shift operation is performed, the DDRAM address shifts.

Display position	1	2	3	4	5	6	7	8
DDRAM address	00	01	02	03	04	05	06	07

For shift left	01	02	03	04	05	06	07	08
----------------	----	----	----	----	----	----	----	----

For shift right	4F	00	01	02	03	04	05	06
-----------------	----	----	----	----	----	----	----	----

### 1-Line by 8-Character Display Example

Display position	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
DDRAM address	00	01	02	03	04	05	06	07	08	09	0A	0B	0C	0D	0E	0F
	40	41	42	43	44	45	46	47	48	49	4A	4B	4C	4D	4E	4F

	HD44780U display								Extension driver display							
--	------------------	--	--	--	--	--	--	--	--------------------------	--	--	--	--	--	--	--


For shift left	01	02	03	04	05	06	07	08	09	0A	0B	0C	0D	0E	0F	10
	41	42	43	44	45	46	47	48	49	4A	4B	4C	4D	4E	4F	50

For shift right	27	00	01	02	03	04	05	06	07	08	09	0A	0B	0C	0D	0E
	67	40	41	42	43	44	45	46	47	48	49	4A	4B	4C	4D	4E

### 2-Line by 16-Character Display Example

Ο ρόλος των μνημών RAM και ROM στην απεικόνιση δεδομένων στην οθόνη αποδίδεται από το παρακάτω σχεδιάγραμμα.



### Καταχωρητές

Ο ελεγκτής HD44780 έχει δύο καταχωρητές των 8 bit που μπορούν να προσπελαστούν άμεσα από το χρήστη. Τον καταχωρητή εντολών (IR) και τον καταχωρητή δεδομένων (DR) όπως φαίνεται και στο Σχ. 3.4.

Ο καταχωρητής εντολών αποθηκεύει κωδικούς εντολών και διευθύνσεις για τις μνήμες RAM του ελεγκτή.

Ο καταχωρητής δεδομένων αποθηκεύει προσωρινά τα δεδομένα που πρόκειται να αποθηκευτούν στις RAM του ελεγκτή ή δεδομένα που διαβάστηκαν από αυτές. Δεδομένα που αποθηκεύονται στον καταχωρητή δεδομένων μεταφέρονται αυτόματα στην κατάλληλη RAM. Όταν ο καταχωρητής εντολών δεχτεί πληροφορίες διεύθυνσης, δεδομένα αποθηκεύονται στον καταχωρητή δεδομένων από την κατάλληλη RAM αυτόματα.

Στο μετρητή διευθύνσεων βρίσκεται αποθηκευμένη η τρέχουσα διεύθυνση της RAM απεικόνισης δεδομένων ή της RAM δημιουργίας χαρακτήρων. Το περιεχόμενό του μπορεί να τροποποιηθεί με μια από τις εντολές που αλλάζουν την διεύθυνση κάποιας από τις RAM του ελεγκτή. Το ποια μνήμη επιλέγεται εξαρτάται από την εντολή. Η διεύθυνση που βρίσκεται αποθηκευμένη σε αυτόν τον καταχωρητή αυξάνεται κατά 1 (ή μειώνεται κατά 1) αυτόματα με κάθε ανάγνωση/εγγραφή από τη μνήμη.


Η διεπαφή με την οποία κάποιος ανταλλάσσει εντολές με τον ελεγκτή φαίνεται στον παρακάτω πίνακα.

Pin No.	Symbol	Level	Description
1	V <sub>SS</sub>	0V	Ground
2	V <sub>DD</sub>	5.0V	Supply Voltage for logic
3	VO	(Variable)	Operating voltage for LCD
4	RS	H/L	H: DATA, L: Instruction code
5	R/W	H/L	H: Read(MPU→Module) L: Write(MPU→Module)
6	E	H,H→L	Chip enable signal
7	DB0	H/L	Data bus line
8	DB1	H/L	Data bus line
9	DB2	H/L	Data bus line
10	DB3	H/L	Data bus line
11	DB4	H/L	Data bus line
12	DB5	H/L	Data bus line
13	DB6	H/L	Data bus line
14	DB7	H/L	Data bus line
15	A	—	LED +
16	K	—	LED —

Ο ελεγκτής HD44780 μπορεί να δεχτεί εντολές είτε σε μια ενιαία εντολή των 8 bit είτε σε μια εντολή που αποστέλλεται σε δύο διαδοχικά τμήματα των 4 bit. Στο αναπτυξιακό δεν μπορούμε να στείλουμε εντολές στον ελεγκτή σε μορφή 8 bit, συνεπώς δεν θα ασχοληθούμε άλλο με την μεταφορά εντολών σε αυτή τη μορφή.

Όταν οι εντολές μεταφέρονται σε δύο τμήματα των 4 bit χρησιμοποιούνται μόνο οι γραμμές DB7 – DB4. Η χρονική ακολουθία λειτουργιών φαίνεται στο Σχ.3.5. Τα 4 περισσότερο σημαντικά bit πρέπει να μεταφερθούν πρώτα. Επειδή δεν υπάρχει δυνατότητα ανάγνωσης της σημαίας BUSY FLAG για να γνωρίζουμε πότε ο ελεγκτής είναι έτοιμος να δεχτεί νέα εντολή πρέπει να εισάγουμε καθυστέρηση μεταξύ των διαδοχικών

εντολών. Ο χρόνος που χρειάζεται ο ελεγκτής HD44780 για να εκτελέσει κάθε εντολή είναι μεταβλητός και εξαρτάται από την εντολή.



**Σχήμα 3.5.** Χρονική ακολουθία λειτουργιών του ελεγκτή HD44780: εγγραφή εντολών, ανάγνωση διευθύνσεων ή δεδομένων.

Το σύνολο εντολών του ελεγκτή περιέχεται στον ακόλουθο πίνακα.

Instruction	Instruction Code										Description	Execution time (fosc=270Khz)
	RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0		
Clear Display	0	0	0	0	0	0	0	0	0	1	Write "00H" to DDRAM and set DDRAM address to "00H" from AC	1.53ms
Return Home	0	0	0	0	0	0	0	0	1	—	Set DDRAM address to "00H" from AC and return cursor to its original position if shifted. The contents of DDRAM are not changed.	1.53ms
Entry Mode Set	0	0	0	0	0	0	0	1	I/D	SH	Assign cursor moving direction and enable the shift of entire display.	39 $\mu$ s
Display ON/OFF Control	0	0	0	0	0	0	1	D	C	B	Set display (D), cursor (C), and blinking of cursor (B) on/off control bit.	39 $\mu$ s
Cursor or Display Shift	0	0	0	0	0	1	S/C	R/L	—	—	Set cursor moving and display shift control bit, and the direction, without changing of DDRAM data.	39 $\mu$ s
Function Set	0	0	0	0	1	DL	N	F	—	—	Set interface data length (DL:8-bit/4-bit), numbers of display line (N:2-line/1-line)and, display font type (F:5x11 dots/5x8 dots)	39 $\mu$ s
Set CGRAM Address	0	0	0	1	AC5	AC4	AC3	AC2	AC1	AC0	Set CGRAM address in address counter.	39 $\mu$ s
Set DDRAM Address	0	0	1	AC6	AC5	AC4	AC3	AC2	AC1	AC0	Set DDRAM address in address counter.	39 $\mu$ s
Read Busy Flag and Address	0	1	BF	AC6	AC5	AC4	AC3	AC2	AC1	AC0	Whether during internal operation or not can be known by reading BF. The contents of address counter can also be read.	0 $\mu$ s
Write Data to RAM	1	0	D7	D6	D5	D4	D3	D2	D1	D0	Write data into internal RAM (DDRAM/CGRAM).	43 $\mu$ s
Read Data from RAM	1	1	D7	D6	D5	D4	D3	D2	D1	D0	Read data from internal RAM (DDRAM/CGRAM).	43 $\mu$ s

\* "—" : don't care


Τέλος για να γίνει δεκτή μια εντολή από τον ελεγκτή πρέπει να τηρούνται οι χρόνοι του παρακάτω πίνακα.

$T_a=25^{\circ}\text{C}$ ,  $V_{DD}=5.0\text{V}$

Item	Symbol	Min	Typ	Max	Unit
Enable cycle time	$T_C$	1200	—	—	ns
Enable pulse width	$T_{PW}$	140	—	—	ns
Enable rise/fall time	$T_R, T_F$	—	—	25	ns
Address set-up time (RS, R/W to E)	$t_{AS}$	0	—	—	ns
Address hold time	$t_{AH}$	10	—	—	ns
Data set-up time	$t_{DSW}$	40	—	—	ns
Data hold time	$t_H$	10	—	—	ns

**Χρήση της οθόνης LCD του αναπτυξιακού easyAVR6.**

Η συνδεσμολογία της οθόνης lcd με τον μικροελεγκτή φαίνεται στο παρακάτω σχήμα, απ' όπου μπορούμε να συμπεράνουμε ότι οι ακροδέκτες PD7 – PD2 πρέπει να είναι ρυθμισμένοι για έξοδο.



Αρχικά χρειαζόμαστε μια ρουτίνα που θα μεταφέρει τα δύο τμήματα των 4 bit κάθε εντολής. Η ρουτίνα θα πρέπει να αφήνει ανεπηρέαστους τους ακροδέκτες που επιλέγουν μεταξύ καταχωρητή εντολών και καταχωρητή δεδομένων, ώστε να μπορεί να χρησιμοποιηθεί και για τις δύο λειτουργίες.

**Ρουτίνα: write\_2\_nibbles**

Αποστολή ενός byte, 4 bit τη φορά στον ελεγκτή της οθόνης lcd. Το λογικό επίπεδο που βρίσκεται ο ακροδέκτης που αντιστοιχεί στο σήμα R/S δεν επηρεάζεται.

**Είσοδος:** Το byte που μεταδίδεται είναι αποθηκευμένο στον καταχωρητή r24

**Έξοδος:** -

**Καταχωρητές:** r25:r24

**Καλούμενες υπορουτίνες:** -

write\_2\_nibbles:

```

push r24                ; στέλνει τα 4 MSB
in r25 ,PIND             ; διαβάζονται τα 4 LSB και τα ξαναστέλνουμε
andi r24 ,0x0f           ; για να μην χαλάσουμε την όποια προηγούμενη κατάσταση
andi r24 ,0xf0           ; απομονώνονται τα 4 MSB και
add r24 ,r25             ; συνδυάζονται με τα προϋπάρχοντα 4 LSB
out PORTD ,r24           ; και δίνονται στην έξοδο
sbi PORTD ,PD3           ; δημιουργείται παλμός Enable στον ακροδέκτη PD3
cbi PORTD ,PD3           ; PD3=1 και μετά PD3=0
pop r24                 ; στέλνει τα 4 LSB. Ανακτάται το byte.
swap r24                ; εναλλάσσονται τα 4 MSB με τα 4 LSB
andi r24 ,0xf0           ; που με την σειρά τους αποστέλλονται
add r24 ,r25
out PORTD ,r24
sbi PORTD ,PD3           ; Νέος παλμός Enable
cbi PORTD ,PD3
ret

```

Στη συνέχεια, με βάση την προηγούμενη ρουτίνα μπορούμε να δημιουργήσουμε δύο άλλες. Η μία θα στέλνει εντολές στην οθόνη και η άλλη δεδομένα.

**Ρουτίνα: lcd\_data**

Αποστολή ενός byte δεδομένων στον ελεγκτή της οθόνης lcd. Ο ελεγκτής πρέπει να βρίσκεται σε 4 bit mode.

**Είσοδος:** Το byte που μεταδίδεται είναι αποθηκευμένο στον καταχωρητή r24

**Καταχωρητές:** r25:r24

**Καλούμενες υπορουτίνες:** wait\_usec, write\_2\_nibbles

lcd\_data:

```
sbi PORTD ,PD2      ; επιλογή του καταχωρητή δεδομένων (PD2=1)
rcall write_2_nibbles ; αποστολή του byte
ldi r24 ,43          ; αναμονή 43μsec μέχρι να ολοκληρωθεί η λήψη
ldi r25 ,0            ; των δεδομένων από τον ελεγκτή της lcd
rcall wait_usec
ret
```

**Ρουτίνα: lcd\_command**

Αποστολή μιας εντολής στον ελεγκτή της οθόνης lcd. Ο ελεγκτής πρέπει να βρίσκεται σε 4 bit mode.

**Είσοδος:** Η εντολή που μεταδίδεται είναι αποθηκευμένη στον καταχωρητή r24

**Έξοδος:** -

**Καταχωρητές:** r25:r24

**Καλούμενες υπορουτίνες:** wait\_usec, write\_2\_nibbles

lcd\_command:

```
cbi PORTD ,PD2      ; επιλογή του καταχωρητή εντολών (PD2=1)
rcall write_2_nibbles ; αποστολή της εντολής και αναμονή 39μsec
ldi r24 ,39          ; για την ολοκλήρωση της εκτέλεσης της από τον ελεγκτή της lcd.
ldi r25 ,0            ; ΣΗΜ.: υπάρχουν δύο εντολές, οι clear display και return home,
rcall wait_usec      ; που απαιτούν σημαντικά μεγαλύτερο χρονικό διάστημα.
ret
```

Τώρα που μπορούμε να στείλουμε εντολές και δεδομένα στην οθόνη μένει να την αρχικοποιήσουμε στην επιθυμητή κατάσταση, ώστε να μπορεί να χρησιμοποιηθεί. Όταν η οθόνη τροφοδοτείται με ρεύμα για πρώτη φορά ο ελεγκτής HD44780 πραγματοποιεί μια εσωτερική αρχικοποίηση και για αυτό απαιτείται να περιμένουμε 40 ms. Στη συνέχεια ο ελεγκτής βρίσκεται σε 8 bit mode και είναι έτοιμος να λάβει εντολές. Ο κώδικας που θα κάνει την αρχικοποίηση δεν πρέπει να βασίζεται στο ότι ο ελεγκτής βρίσκεται σε 8 bit mode, διότι αυτό δεν είναι πάντα αληθές. Κάθε φορά που προγραμματίζουμε τον μικροελεγκτή αυτός ξεκινάει την εκτέλεση του κώδικα από την αρχή, η οθόνη όμως βρίσκεται στην κατάσταση που την αφήσαμε την προηγούμενη φορά. Για να οδηγήσουμε την οθόνη σε 4 bit mode στέλνουμε δύο φορές την εντολή 0x30 (function set) για 8 bit mode. Η συγκεκριμένη εντολή (μεταξύ άλλων) ρυθμίζει τον ελεγκτή να δέχεται εντολές και δεδομένα σε ένα ενιαίο κομμάτι των 8 bit. Τα 4 λιγότερο σημαντικά bit μας είναι αδιάφορα. Αν ο ελεγκτής είναι σε 8 bit mode δεν θα αλλάξει κάτι, αν όμως είναι σε 4 bit mode θα μεταβεί σε 8 bit mode. Μόλις είμαστε βέβαιοι για την μορφή που πρέπει να στέλνουμε τις εντολές μπορούμε να προχωρήσουμε με την αρχικοποίηση.

**Ρουτίνα: lcd\_init**

Αρχικοποίηση και ρυθμίσεις της οθόνης LCD όπως παρουσιάζεται παρακάτω:

DL = 0 4 bit mode

N = 1 2 lines

F = 0 5×8 dots

D = 1 display on

C = 0 cursor off

B = 0 blinking off

I/D = 1 ddram address auto increment

SH = 0 shift of entire display off

**Είσοδος:** -

**Έξοδος:** -

**Καταχωρητές:** r25:r24



**Καλούμενες υπορουτίνες:** wait\_msec, wait\_usec, lcd\_command

lcd\_init:

```

ldi r24 ,40          ; Όταν ο ελεγκτής της lcd τροφοδοτείται με
ldi r25 ,0           ; ρεύμα εκτελεί την δική του αρχικοποίηση.
rcall wait_msec      ; Αναμονή 40 msec μέχρι αυτή να ολοκληρωθεί.

ldi r24 ,0x30        ; εντολή μετάβασης σε 8 bit mode
out PORTD ,r24       ; επειδή δεν μπορούμε να είμαστε βέβαιοι
sbi PORTD ,PD3       ; για τη διαμόρφωση εισόδου του ελεγκτή
cbi PORTD ,PD3       ; της οθόνης, η εντολή αποστέλλεται δύο φορές
ldi r24 ,39
ldi r25 ,0           ; εάν ο ελεγκτής της οθόνης βρίσκεται σε 8-bit mode
rcall wait_usec      ; δεν θα συμβεί τίποτα, αλλά αν ο ελεγκτής έχει διαμόρφωση
                    ; εισόδου 4 bit θα μεταβεί σε διαμόρφωση 8 bit

ldi r24 ,0x30
out PORTD ,r24
sbi PORTD ,PD3
cbi PORTD ,PD3
ldi r24 ,39
ldi r25 ,0
rcall wait_usec

ldi r24 ,0x20        ; αλλαγή σε 4-bit mode
out PORTD ,r24
sbi PORTD ,PD3
cbi PORTD ,PD3
ldi r24 ,39
ldi r25 ,0
rcall wait_usec

ldi r24 ,0x28        ; επιλογή χαρακτήρων μεγέθους 5x8 κουκίδων
rcall lcd_command    ; και εμφάνιση δύο γραμμών στην οθόνη

ldi r24 ,0x0c        ; ενεργοποίηση της οθόνης, απόκρυψη του κέρσορα
rcall lcd_command

ldi r24 ,0x01        ; καθαρισμός της οθόνης
rcall lcd_command
ldi r24 ,low(1530)
ldi r25 ,high(1530)
rcall wait_usec


ldi r24 ,0x06        ; ενεργοποίηση αυτόματης αύξησης κατά 1 της διεύθυνσης
rcall lcd_command    ; που είναι αποθηκευμένη στον μετρητή διευθύνσεων και
                    ; απενεργοποίηση της ολίσθησης ολόκληρης της οθόνης

ret

```

### Τα ζητούμενα της 3<sup>ης</sup> εργαστηριακής άσκησης του AVR

**Ζήτημα 3.1** Να εξομοιωθεί η λειτουργία ενός υποθετικού I.C. με πύλες όπως φαίνεται στο σχήμα 3.6. Τα bits εισόδου πρέπει να αντιστοιχούν ακριβώς όπως φαίνονται στο σχήμα 1 με τα dip switches PA7-0 και οι έξοδοι πρέπει να είναι τα leds PB3-0. Οι πύλες 1, 2, 3, 4, 5 να υλοποιούν τις AND, AND, NOR, NXOR και OR αντίστοιχα. Όταν πατάμε κάποιο από τα push buttons PC7-0 να αντιστρέφεται η ένδειξη του αντίστοιχου led PA7-0. Αρχικά όλα τα led να είναι σβηστά και όλες οι αλλαγές να γίνονται κατά το πάτημα των push buttons. Το πρόγραμμα να δοθεί σε assembly.



Σχήμα 3.6 Υποθετικό I.C. με πύλες

**Ζήτημα 3.2** Να υλοποιηθούν σε ένα σύστημα μικροελεγκτή AVR οι λογικές συναρτήσεις:

$$F0 = (AB + BC + CD + DE)', \quad F1 = ABCD + EF', \quad F2 = F0 + F1$$

και να εμφανιστούν στα τρία LSB της θύρας εξόδου PortA (0-2). Οι μεταβλητές εισόδου (A, B, C, D, E) υποθέτουμε ότι αντιστοιχούν στα bit της θύρας εισόδου PortC (0-4). Το πρόγραμμα να δοθεί σε C.

**Ζήτημα 3.3** Γράψτε ένα πρόγραμμα το οποίο ανάβει τα leds PA0-7 για 2 sec μόνο όταν πατηθούν στη σειρά δύο συγκεκριμένα πλήκτρα στο keypad 4×4 (π.χ. ο διψήφιος αριθμός της ομάδας σας). Ανεξάρτητα από το χρονικό διάστημα για το οποίο θα μείνει πατημένο ένα πλήκτρο, το πρόγραμμά σας θα πρέπει να θεωρεί ότι πατήθηκε μόνο μια φορά. Το πρόγραμμα αυτό θα μπορούσε να αποτελέσει τη βάση μιας «ηλεκτρονικής κλειδαριάς». Ξεκινήστε σχεδιάζοντας απαραίτητα το διάγραμμα ροής και μετά γράψτε το πρόγραμμα.


**Ζήτημα 3.4** Γράψτε ένα πρόγραμμα που να απεικονίζει στην οθόνη LCD το πλήκτρο (χαρακτήρα) που πατήθηκε τελευταίο στο keypad 4×4. Στο ξεκίνημα, μέχρι να πατηθεί κάποιο πλήκτρο, η οθόνη να εμφανίζει την ένδειξη TEAM XX (τον αριθμό της ομάδας σας). Τόσο η ένδειξη TEAM XX όσο και ο χαρακτήρας που πατήθηκε να εμφανίζονται στην πάνω αριστερή θέση της οθόνης. Ο κέρσορας να μην είναι ορατός.

**ΠΡΟΑΙΡΕΤΙΚΕΣ ΑΣΚΗΣΕΙΣ** (bonus 0.5 στο βαθμό του εργαστηρίου)

**Ζήτημα 3.5** Γράψτε ένα πρόγραμμα που να εξομοιώνει ένα σύστημα συναγερμού. Υποθέτουμε ότι τα push buttons PB0-PB7 αντιστοιχούν στις εξόδους των αισθητήρων. Αν κάποιο από αυτά πατηθεί τότε μέσα σε 10 sec πρέπει να πληκτρολογηθεί στο keypad ο αριθμός της ομάδας (δύο ψηφία) και ένας διψήφιος ακόμα αριθμός. Αν είναι σωστά, να εμφανίζεται το μήνυμα «ALARM OFF» στο lcd display. Αλλιώς τίθεται ο συναγερμός με το άναμμα των leds PA0-PA7 που υποθέτουμε ότι αντιστοιχεί στην ενεργοποίηση της σειρήνας. Ταυτόχρονα να εμφανίζεται στο lcd display το μήνυμα «ALARM ON». Κατά την εισαγωγή του κωδικού να εμφανίζονται τα πλήκτρα που πατάει ο χρήστης στο lcd display καθώς και ο κέρσορας. Όλα τα μηνύματα να ξεκινούν από την πάνω αριστερή θέση του display, ενώ το υπόλοιπο να είναι κενό.

**Ζήτημα 3.6** Γράψτε πρόγραμμα που να απεικονίζει στο lcd display την δυαδική τιμή σε συμπλήρωμα ως προς 2 που διαβάζει από την θύρα PORTA σε δεκαδική μορφή τριών ψηφίων με το πρόσημο (δηλαδή αν από τη θύρα διαβαστεί π.χ. ο αριθμός 0110 0001 =  $97_{10}$ , τότε ξεκινώντας από την πάνω αριστερή θέση του display να εμφανιστεί ο αριθμός +97 ενώ αν δοθεί ο 1000 0000 να εμφανίζεται αντίστοιχα ο αριθμός -128). Η διαδικασία να είναι συνεχόμενη.

Δίνεται στο Σχήμα 3.7 ένα διάγραμμα ροής για τη μετατροπή ενός προσημασμένου δυαδικού αριθμού σε BCD μορφή που αν θέλετε μπορείτε να ακολουθήσετε.



**Σχήμα 3.7** Διάγραμμα ροής για τη μετατροπή προσημασμένου δυαδικού αριθμού σε BCD μορφή